interval [-π,π]

Square Wave Fourier Transform

In[95]:=
```
Clear["Global`*"]
```

In[96]:=
```
c[n_] = 1/(2 π) Integrate[Exp[- I n x] 1, {x, 0, 1 π}] // Simplify
```

Out[96]=
$$-\frac{i\left(1-e^{-i\,n\,\pi}\right)}{2\,n\,\pi}$$

In[97]:=
```
c[0] = 1/(2 π) Integrate[1, {x, 0, 1 π}]
```

Out[97]=
$$\frac{1}{2}$$

In[98]:=
```
? c
```

Out[98]=

Symbol

Global`c

Definitions

$c[0] = \frac{1}{2}$

$c[n\_] = -\frac{i\left(1-e^{-i\,n\,\pi}\right)}{2\,n\,\pi}$

Full Name Global`c

In[99]:=
```
term[x_, n_] = c[n] Exp[I n x]
```

Out[99]=
$$-\frac{i\,e^{i\,n\,x}\left(1-e^{-i\,n\,\pi}\right)}{2\,n\,\pi}$$

In[100]:=
```
term[x_, 0] = c[0]
```

Out[100]=
$$\frac{1}{2}$$

In[101]:=
```
term[x, n] // (Simplify[#, Assumptions → n ∈ Integers]) &
```

Out[101]=
$$\frac{i\left(-1+(-1)^{n}\right)e^{i\,n\,x}}{2\,n\,\pi}$$

In[102]:=
```
term[x, 5]
```

Out[102]=
$$-\frac{i\,e^{5\,i\,x}}{5\,\pi}$$

In[103]:=
```
term[x, 3] // ExpToTrig
```

Out[103]=
$$-\frac{i\,\text{Cos}[3\,x]}{3\,\pi}+\frac{\text{Sin}[3\,x]}{3\,\pi}$$

In[104]:=
```
Clear[sum];
sum[x_, k_] := Sum[term[x, n], {n, -k, k}]
```

In[106]:=
```
sum[x, 3]
```

Out[106]=
$$\frac{1}{2}+\frac{i\,e^{-i\,x}}{\pi}-\frac{i\,e^{i\,x}}{\pi}+\frac{i\,e^{-3\,i\,x}}{3\,\pi}-\frac{i\,e^{3\,i\,x}}{3\,\pi}$$

In[107]:=
```
sum[x, 3] // ExpToTrig
```

Out[107]=
$$\frac{1}{2}+\frac{2\,\text{Sin}[x]}{\pi}+\frac{2\,\text{Sin}[3\,x]}{3\,\pi}$$

In[108]:=
```
ticks = Table[n Pi/2, {n, -100, 100, 1}];
```

In[109]:=
```
p1 = Plot[sum[x, 20] // Re , {x, -2 Pi, 2 Pi},
  GridLines → Automatic ,
  Ticks → {ticks,  Automatic},
  PlotRange → Automatic]
```

Out[109]=



In[110]:=
```
p1 = Plot[sum[x, 100] // Re , {x, -2 π, 2 π}, GridLines → Automatic]
```

Out[110]=



In[111]:=
```
p2 = Plot[1 + Floor[Sin[x]], {x, -2 π, 2 π}, PlotStyle →  {Red, Thick}]
```

Out[111]=



In[112]:=
```
Show[p1, p2]
```

Out[112]=



Sawtooth Wave Fourier Transform

In[113]:=
```
Clear["Global`*"]
```

In[114]:=
```
c[n_] = 1/(2 π) Integrate[Exp[- I n x] x, {x, -π, π}] // Simplify
```

Out[114]=
$$\frac{i\,(n\,\pi\,\text{Cos}[n\,\pi] - \text{Sin}[n\,\pi])}{n^2\,\pi}$$

In[115]:=
```
c[0] = 1/(2 π) Integrate[ x , {x, -π, π}]
```

Out[115]= 0

In[116]:=
```
? c
```

Out[116]=

Symbol

Global`c

Definitions	c[0] = 0
	$c[n\_] = \frac{i\,(n\,\pi\,\text{Cos}[n\,\pi] - \text{Sin}[n\,\pi])}{n^2\,\pi}$
Full Name	Global`c

In[117]:=
```
term[x_, n_] = c[n] Exp[I n x]
```

Out[117]=
$$\frac{i\,e^{i\,n\,x}\,(n\,\pi\,\text{Cos}[n\,\pi] - \text{Sin}[n\,\pi])}{n^2\,\pi}$$

In[118]:=
```
term[x_, 0] = c[0]
```

Out[118]= 0

In[119]:=
```
term[x, n] // (Simplify[#, Assumptions → n ∈ Integers]) &
```

Out[119]=
$$\frac{i\,(-1)^n\,e^{i\,n\,x}}{n}$$

In[120]:=
```
term[x, 5]
```

Out[120]=
$$-\frac{1}{5}\,i\,e^{5\,i\,x}$$

In[121]:=
```
term[x, 3] // ExpToTrig
```

Out[121]=
$$-\frac{1}{3}\,i\,\text{Cos}[3\,x] + \frac{1}{3}\,\text{Sin}[3\,x]$$

In[122]:=
```
Clear[sum];
sum[x_, k_] := Sum[term[x, n], {n, -k, k}]
```

In[124]:=
```
sum[x, 3]
```

Out[124]=
$$i\,e^{-i\,x} - i\,e^{i\,x} - \frac{1}{2}\,i\,e^{-2\,i\,x} + \frac{1}{2}\,i\,e^{2\,i\,x} + \frac{1}{3}\,i\,e^{-3\,i\,x} - \frac{1}{3}\,i\,e^{3\,i\,x}$$

In[125]:=
```
sum[x, 3] // ExpToTrig
```

Out[125]=
$$2\,\text{Sin}[x] - \text{Sin}[2\,x] + \frac{2}{3}\,\text{Sin}[3\,x]$$

In[126]:=
```
ticks = Table[n Pi/2, {n, -100, 100, 1}];
```

In[127]:=
```
p1 = Plot[sum[x, 20] // Re , {x, -2 π, 2 π},
  GridLines → Automatic ,
  Ticks →  {ticks,  Automatic},
  PlotRange → Automatic]
```

Out[127]=



In[128]:=
```
p1 = Plot[sum[x, 100] // Re , {x, -8 π, 8 π}, GridLines → Automatic]
```

Out[128]=



In[129]:=
$$p2 = \text{Plot}\left[x - 2\,\pi\,\text{Floor}\left[\frac{x+\pi}{2\,\pi}\right], \{x, -8\,\pi, 8\,\pi\}, \text{PlotStyle} \to \{\text{Red}, \text{Thick}\}\right]$$

Out[129]=



In[130]:=
```
Show[p1, p2]
```

Out[130]=